Lab Practical - Discontinuum Analysis & Distinct Element Method

Part A – The Basics

The Universal Distinct Element Code (UDEC) is a two-dimensional numerical program based on the distinct element method for discontinuum modeling. UDEC simulates the response of discontinuous media (such as a jointed rock mass) subjected to either static or dynamic loading. In UDEC, the construction of a numerical model generally follows the same basic steps, as illustrated to the right.

This first example problem presents a simple example to demonstrate some of the basic aspects of solving problems with UDEC. The example is a four-block slope stability problem. We evaluate the stability for different conditions of joint friction assuming rigid block behaviour (i.e. non-deformable blocks).



1) To begin, load UDEC by running the UDEC executable. Your computer will load the program and display an initial heading followed by the interactive prompt - udec>.

   The latest version of UDEC has a graphical interface, however, it is more instructive to work from the command line and viewing the results directly.

2) Begin by defining the problem geometry starting with a single block using the BLOCK command. Type the following line at the command prompt:

   block (0,0) (0,20) (20,20) (20,0)

   This command creates a square block with side lengths of 20 units (in this case, meters). To see the block, type:

   plot block

   A picture of the block will appear on the screen. Note that the corners appear slightly rounded. The rounding is used in UDEC to give reasonable physical behaviour to blocks which just slightly overlap each other. The rounding length may be adjusted by the user. To continue the problem, hit the <Enter> key.

3) The problem is continued by splitting the initial block into smaller blocks by typing:

   crack (0,2) (20,8)
   crack (5,3) (5,20)
   crack (5,12) (20,18)

   These commands split the initial block along lines with endpoints specified by the x,y coordinates in parentheses. To see the resultant problem geometry, again type:

   plot block



4) Next, the lower-most and left-most blocks are immobilized by typing:

   fix range 0,20 0,5
   fix range 0,5 0,20

   This command fixes the velocity (i.e., zero) of all blocks with centroids in the range $0<x<20$, $0<y<5$ and $0<x<5$, $0<y<20$. Try to visualize which blocks in your model have been fixed by this command.

5) Required material properties are assigned for the blocks and joints by typing:

   prop mat=1 dens=2000
   prop jmat=1 jkn=1.33e7 jks=1.33e7 jfric=20.0

   For this problem, the mass density of all blocks is specified to be 2000 kg/m^3. Note that the mass density is assigned, not the unit weight of the block material. All joints are specified to have contact normal (jkn) and shear (jks) stiffness equal to 1.33×10^7 Pa/m and friction angles equal to 20º. As will be seen later, different properties can be assigned to various joints and intact blocks.

6) Next, gravitational accelerations in x- and y-directions are specified by typing:

   set gravity 0 -10

   In order to absorb vibrational energy, damping is introduced by typing:

   damp local

   This is the default damping condition in UDEC, and the command DAMP local is actually not required here. We do so only to emphasize that this is a static analysis.

7) At this point, the problem is ready to be executed. As will be seen later, it is often helpful to judge behaviour (i.e., equilibrium, stability, instability) by observing the motion of specified points in the rock mass. In this problem, we monitor the y-velocity of a point at the top right corner of the model. Type:

   hist yvel 20,20

   Following execution of this command, the program will return information about the selected monitoring point (20,20).

8) One hundred calculation cycles are executed by typing:

   step 100

   During execution, the current cycle count and maximum out-of-balance force are printed on the screen every 10 cycles. Inspection of these values indicates that equilibrium has been obtained (the out-of-balance force approaches zero). A graphical representation of this behaviour is obtained by typing:

   plot hist 1

   Next, again type:

   plot block

9) It is often helpful to save this initial state so that it can be restarted at any time — for example, to perform parameter studies. To save the current state (in a file called "SLOPE.SAV"), type:

   save slope.sav

   Note that if you wish to save this file to a specific directory, you must include the path name with the file name (e.g. *save d:\yourdirectory\slope.sav*).

10) The behavior of the slope can now be studied by removing the left-most block by typing:

   delete range 0,5 0,20

   This command deletes blocks with centroids in the range 0<x<5, 0<y<20. Check this by typing:

   plot block

11) Next, the calculation process continues:

   cycle 1000
   plot block velocity

   The problem state after 1000 additional cycles (1100 cycles total) should show that only the top block is sliding. This is the expected result because the friction angle (20º) is less than the slope (22º) of the joint between the two upper-most blocks.

   A plot of the y-velocity history of the monitored point obtained by typing:

   plot hist 1

   This history indicates that the block is sliding; the velocity increases at a constant rate.

12) The problem may be continued in the manner previously described, but it is interesting to examine the effect of other choices of problem parameters at this point. The initial save state may be recalled by typing:

   restore d:\yourdirectory\slope.sav

The left-most block is removed as before but, in this case, the joint friction angle is reduced to 11º. To do so, type:

delete 0,5 0,20
prop jmat=1 jfric=11
cycle 1000
plot block velocity

Now, both blocks are sliding (the top block is sliding faster than the middle block). This is the expected result because the friction angle (11º) is less than the dip of the joints (22º for the top and 17º for the bottom).



Part B – Deformable Blocks

This example will demonstrate how to model a discontinuous rock slope using deformable blocks. In this case we will examine the instability mechanism referred to as flexural toppling.

1) For this tutorial problem, we will type our commands into a text editor and then read the file into the program.

   To begin, open a text editor (e.g. notepad) and load UDEC by running the UDEC executable.

2) The problem set-up begins by defining the problem geometry starting with a single block using the BLOCK command. Within the text editor, enter:

   new
   round 1
   block 0,0 0,1000 600,1000 800,400 1200,400 1200,0

   The rounding command rounds the block corners to prevent unnatural stress concentrations. In nature, such sharp corners are typically crushed allowing the block to assume rounded corners naturally.

3) Next we enter the jointing pattern for the rock mass. Add the following commands:

   jregion id=1 0,0 20,1000 800,1000 800,0
   jset 75,0 1000,0 0,0 30,0 0,0 range jreg=1
   jdelete

   The 'jregion' command defines the region that will be jointed and the 'jset' command defines the jointing pattern (dip angle, persistence, gap length, spacing and location of first joint). Since UDEC cannot compute problems where the blocks are only partially divided, the 'jdelete' command removes any non-through-going joints.

Before moving on, check the problem geometry to make sure everything is correct. Save the text edit file with the extension .dat and then switch to UDEC. At the UDEC command prompt type:

call <filename.dat>

You should see your commands from the text edit file scroll past. When finished, type at the UDEC prompt

plot block



4) Simply through visual inspection, it should be obvious that if the blocks were rigid and no cross-cutting joints were present, the slope would be kinematically stable. In other words, the slope could not fail as long as the blocks remain non-deformable. In nature though, rock does deform. To account for block deformation, we need to discretize the blocks with a finite-difference mesh. Returning to the text editor, add the command:

gen edge 200

This command automatically generates a mesh of triangular elements with edge lengths of approximately 200 m. To see the mesh generated, save the text editor file and switch to the UDEC window.

call <filename.dat>
plot block zone

5) Now set properties for the deformable blocks. In your text editor, add:

prop mat=1 dens=2700 k=20e9 g=12e9
prop jmat=1 jkn=1e10 jks=1e9 jfric=40.0 jcoh=5e6

The first command sets the elastic properties for the block (in terms of a bulk and shear modulus). The second command enters the properties for the joints.

6) Next define the boundary conditions. For this problem, fix the x-velocities on the left and right boundaries so that only movements in the y-direction are possible. Along the bottom of the model, fix the y-velocity. In your text edit file add the commands:

bound xvel=0 range -1 1 -1 1001
bound xvel=0 range 1199 1201 -1 401
bound yvel=0 range -1 1201 -1 1

Note that the above numbers define a box around the boundary to be fixed via x-lower, x-upper and y-lower, y-upper pairs.

To check to make sure the boundaries are fixed correctly, save your text edit file and read it into UDEC.

call <filename.dat>
plot bl bound xc
plot bl bound yc

7) Next we must set our loading conditions which include our *in situ* stress state. In this case, assuming a rock density of 2700 kg/m3 and a horizontal to vertical stress ratio of k=1.3.

insitu stress -35e6 0 -27e6 ygrad 3.5e4 0 2.7e4 &
szz -35e6 zgrad 0 3.5e4

grav 0 -10

Enter the above lines into your text edit file.

8) We are now ready to solve for our initial state. In your text editor add the following commands:

step 5000
save <filename.sav>

Save and call the file into UDEC.

call <filename.dat>

The problem will then solve to equilibrium and save the solved state. We can check our equilibrium condition by plotting:

pl bl syy fill

This command will plot the initial vertical stresses. These should approximately follow parallel topography.

Now plot the x-displacements:

pl bl xd fill

From this we can see that even assuming elastic behaviour, our joint bound blocks are starting to flex and bend out of the slope.



9) Since we are only interested in the slope behaviour following the initialization of the equilibrium state, we can reset the displacements that accumulated during the solution of the initial state. Returning to the text editor, add:

reset disp jdisp

10) Next we want to change the constitutive model governing the material behaviour from elasticity to an elasto-plastic relationship based on a Mohr-Coulomb strength criterion. In this case, assume the intact properties for a schist (E = 11 GPa, ν = 0.20, c = 1 MPa and ϕ = 20º). Add the following lines to your text edit file:

change cons=3 mat=2 range jreg=1

prop mat=2 den=2700 k=8.8e9 g=4.3e9 coh=1e6 fric=20.0 ten=0
prop jmat=2 jkn=1e10 jks=1e8 jfric=10.0 jcoh=0

These commands re-designate the constitutive model applied to the upper jointed region (originally defined as 'jreg=1") from elastic to elasto-plastic. This may be the case if we have a schist (which generally behaves as a non-elastic material).

11) We can now solve our model to account for elasto-plastic rock deformations. Add the following lines to your text editor, save and call into UDEC.

step 40000
save <filename-2.sav>

12) After the model has finished time-stepping, we can examine our results:

pl bl plas
pl bl xd fill
pl bl mag 100

From these plots we can see that the slope has become unstable, and is failing through flexural toppling. The first plot shows those elements where the material has yielded or failed. In the model we can see what may be a shear plane developing due to the flexing of the blocks.

The second plot shows the horizontal displacements that occur primarily above this shear plane.

The third plot directly shows the flexural yielding of the toppling blocks through a plot that exaggerates the deformation of the blocks. Note the hinge that is developing along these blocks.



Part C – Dynamic Analysis

This example examines the numerical modelling of dynamic loading, such as that experienced during an earthquake or due to blasting. In this case, a granite rock face is modelled for stability before and after it is hit by an earthquake.

It should be noted that dynamic analyses are often very complicated and require a considerable amount of insight to interpret correctly. There are three aspects that the user should consider when preparing a UDEC model for a dynamic analysis. These are: (1) dynamic loading and boundary conditions; (2) mechanical damping; and (3) wave transmission through the model.

The types of dynamic loading and boundary conditions in UDEC are shown to the right:



(a) Flexible base



(a) Rigid base

1) The problem geometry consists of a granite rock slope cut by several jointed blocks. To begin with, a rectangular block is defined from which the problem geometry will be formed. Enter these commands working with a text editor as in the previous example:

```
new
bl  0,0  0,50  80,50  80,0
```

2) A joint set dipping at 76º cuts the block and is cross cut by two shallow dipping joints:

```
jreg  30,10  39.1,50  61,50  52,10
jset  76,0  100,0  0,0  4,0  30,10
crack  0,10  32,10
crack  30,12.5  65,14
crack  35,30  70,31.5
```

3) Save your text edit file and call it into UDEC. Plotting, we see that we only need to remove the upper half of the block to finish the slope geometry.

```
call <filename.dat>
plot block
```

Returning to the text editor, add:

```
delete 0 30 10 50
jdelete
```

Again, save the text edit file and call it into UDEC (as described above) and plot the block geometry.



4) Next, mesh the intact blocks using triangular elements. Returning to the text editor, add:

    gen edge 10

    Confirm this by saving the text edit file, calling it into UDEC and plotting the elements:

    plot block zones

5) For this problem, the rock mass is made up of a strong granite and so we can assume that its intact behaviour will be predominantly elastic. We will also assume that the joints are filled with clay which produces a relatively low joint friction angle.

    prop mat=1 dens=2700 k=45e9 g=30e9
    prop jmat=1 jkn=1e9 jks=1e8 jfric=9

6) Next set the *in situ* stress state and boundary conditions, assuming gravitational loading. In terms of the boundary conditions, velocities normal to the boundary are fixed (i.e. rollers). Add the following lines to the text edit file:

    bound xvel=0 range -1 1 -1 51
    bound xvel=0 range 79 81 -1 51
    bound yvel=0 range -1 81 -1 1

    grav 0 -10

7) Solve for the initial equilibrium conditions by adding the following lines to your text edit file, saving it and reading it into UDEC:

    damp auto
    solve
    save <filename-initial.sav>

8) We can first check our initial state to make sure equilibrium was reached and the slope hasn't failed

    plot block disp yel

    Because we used the "solve" option in the previous step, we know that our model has converged to an approximate equilibrium state. If the slope was unstable, the solve command would continue time-stepping in an attempt to reach equilibrium.



If a sensitivity analysis was performed, we would see that the slope is close to its limit equilibrium state. As a quick test, you may want to try rerunning the previous analysis but using a joint friction angle of 6º (i.e. jfric=6º).

prop jmat=1 jkn=1e9 jks=1e8 jfric=6

If you do so you will see that the slope fails. Now change the joint friction angle back to 9º by re-entering the input lines back from the beginning. Quick parameter sensitivity tests like this demonstrate the usefulness of using a text editor to construct UDEC models. Instead of retyping all the command lines in order to change one property (which would be the case if the commands were directly entered at the UDEC prompt), only those parameters that are being changed need to be retyped in the text editor file.

9) We are now ready to take the initial equilibrium state (which indicates a stable slope), and add our earthquake. But first we need to reset the slope displacements accumulated by initializing the equilibrium state. Add the following line to the text edit file:

reset displ jdis

10) *As stated at the beginning,* there are three aspects that the user must consider when preparing a UDEC model for a dynamic analysis. These are: (1) dynamic loading and boundary conditions; (2) mechanical damping; and (3) wave transmission through the model. The input lines provided below should be added to your text edit file.

For the boundary conditions, we will use viscous (or quiet) boundaries, which mean that when an earthquake wave intersects the model boundary, it will be absorbed instead of reflected.

bound -1 1 -1 51 xvisc
bound 79 81 -1 51 xvisc

The mechanical damping is set as:

damp 0.002 1

Next, to generate our earthquake, the dynamic input can be applied in one of the following ways: a velocity history, a stress history, a force history, or a fluid pressure history (within the joints only).

For this problem, choose the first technique in which a velocity history is prescribed:

bound -1 81 -1 1 xvel 0.5 xhist sine 2 3

This command will generate an earthquake with a frequency of 2 cycles per second lasting 3 seconds long. The magnitude is controlled by the number following 'xvel' in this line.

11) Next add several history points to monitor displacements at various locations within the slope:

hist xvel 65 18
hist xvel 33.5 17
hist xvel 40.5 46.5
hist xvel 58 46.5

To see where these history points are, save the text edit file, call it into UDEC and plot their location using the command:

plot block history

12) Solve for the dynamic event:

cycle 50000

Plotting the results, the model indicates a failed state, which show that the granite blocks have slipped, resembling a reversed-toppling failure.

plot block disp yel



Next plot the velocity histories to see the earthquake wave trace that passed through the model. Start with the following:

plot hist 1

Note that the above command is plotting the "1st" history point that was defined, in this case centered in the unjointed part of the slope (to see location, use '*plot block history*').



Plotting the histories for the other points, we see that failure of the slope occurs as the seismic loading (simulated as an oscillating horizontal velocity i.e. parallel to x-direction) acts to exceed the shear strength along the base of the granite slabs.

13) Return to your text editor and find the line given below:

bound -1 81 -1 1 xvel 0.5 xhist sine 2 3

This line controls the magnitude and duration of the simulated earthquake. The x-velocity ('*xvel*') component controls the magnitude, and the x-history values generate a sine-wave ('*sine*') which controls the frequency and duration of the event (in the example above, the frequency is 2 cycles per second with a duration of 3 seconds).

Systematically reduce the x-velocity value and solve to determine the smallest earthquake permissible for which the slope does not fail.

Part D – Coupled Hydro-Mechanical Analysis

The stability of a slope in jointed rock is affected by the water level behind the slope. In this example, the water level is raised in stages until the slope becomes unstable. The failure of the slope occurs when the fluid pressure in the joints increases (and the effective normal stress in the joints decreases) such that the limiting shear strength of the joints at the slope face is reached.

1) The problem geometry consists of a slope in regularly jointed rock. The water level is raised in four stages to elevations of 6 m, 8 m, 9 m, and 10 m above the slope toe. A steady-state flow analysis is performed at each stage.



Since this is a coupled hydro-mechanical problem, "config fluid" command is required for fluid flow calculation and it must be entered before the "block" command that creates the geometry. Then, the block rounding angle is set and the outline of the problem geometry is defined. In entering the various lines provided in this example, work from a text editor as in the previous examples:

New
config flow
round 0.05
set delc off
block 0,-5 0,0 5,0 11,10 22,10 22,-5

Note that the command SET delc off is specified for this problem. This prevents the contacts behind any failing rock wedges from being deleted. Otherwise, the domains associated with these contacts will become part of the outer domain and the fluid pressure will vanish. It is difficult to determine the actual value for the fluid pressure when large displacement of the wedge occurs. A conservative estimate is to assume that the fluid pressure does not vanish.

2) The two joint sets cutting the block are given as having dip angles of 20º and 80º:

jset 20,0 100,0 0,0 2,0 (5,1)
jset 80,0 100,0 0,0 3,0 (5,0)

and exceptionally small blocks are removed

del area 0.1

Save your text edit file and check that the problem geometry was entered correctly by calling the file into UDEC and plotting the block geometry

call <filename.dat>
plot block



3) Mesh the problem domain using triangular finite-difference elements with edge lengths of 10. Returning to your text edit file add the line:

   gen edge 10

   Confirm this by saving the text edit file and calling the file again into UDEC as described above. Once doing so, plot the block elements.

   plot block zones

4) The mechanical and hydraulic intact block and discontinuity properties are given as:

   Rock Properties:

   density 2500 kg/m^3
   bulk modulus 16.7 GPa
   shear modulus 10.0 GPa

   Joint Mechanical Properties:

   normal stiffness 10 GPa/m
   shear stiffness 10 GPa/m
   friction angle 25º

   Joint Hydraulic Properties:

   permeability factor $1x10^{8}$ MPa^{-1} sec^{-1}
   residual hydraulic aperture $2x10^{-4}$ m
   aperture at zero normal stress $5x10^{-4}$ m

   Fluid Properties:

   density 1000 kg/m^3

   Setting these properties and changing the joint constitutive model to that of a joint area contact elastic/plastic Coulomb slip criterion:

   prop mat=1 dens=2500 k=16.7e9 g=10e9
   prop mat=1 jkn=10e9 jks=10e9 jfric=45.0
   prop mat=1 jperm=1e2 azero=0.0005 ares=0.0002

   change jmat=1 jcons=2

5) Next set the *in situ* stress state and boundary conditions. For the *in situ* stress state, a horizontal to vertical stress ratio of 0.5 (K=0.5) is assumed. In terms of the boundary conditions, velocities normal to the boundary have been fixed (i.e. rollers).

   insitu str -1.25e5 0 -2.5e5 ygrad 1.25e4 0 2.5e4

   bound xvel=0 range -1,1 -6 1
   bound yvel=0 range -1,31 -6 -4
   bound xvel=0 range 21,23 -6 11

6) A history point is chosen located in the upper slope, to monitor the unbalanced forces and displacements

   hist n=100 xdis 11,10 ydis 11,10
   hist unbal

7) Now solve for the initial stress state before introducing water pressures. The slope is brought to an equilibrium state under gravity loading.

    grav 0 -10
    solve
    save slfl1.sav

8) Before continuing, first check the initial state to make sure equilibrium has been reached and then introduce a water table at a height of 6 m

    plot block dis yel
    plot hist 1
    plot hist 2

    The problem is modeled as a steady-state flow analysis by specifying SET flow steady. Enter the following lines in your text edit file:

    fluid dens 1000
    set flow steady

    The water level is raised by changing the fluid pressure gradient for each stage with the BOUND pp pygrad command. The water level at the right-hand side is raised to 6 m above the slope toe; the water level on the left-hand side is maintained at the level of the slope toe.

    bound imperm range -1,31 -6 -4
    bound pygrad -1e4 range -1,1 -6 1
    bound pp=6e4 pygrad -1e4 range 21,23 -6 6

9) The slope displacements accumulated by initializing the equilibrium state are reset, and another history point is chosen located in the slope's toe to monitor the displacements there

    reset displ jdis
    hist xdis 5.92 1.54 ydis 5.92 1.54
    prop mat=1 jfric=25

10) Solving now for the coupled hydro-mechanical condition with the water table set at 6 m, we find that the slope is stable for this fluid pressure condition.

    cyc 500
    save slfl2.sav

    Plotting the groundwater flow vectors and the slope displacement vectors, check whether the flow affects the stability conditions.

    plot bl vf yel ← plots the flow vectors
    plot bl dis yel ← plots the block displacements
    plot bl pp yel ← plots the joint pore pressures

    Make note of (<u>record</u>) the maximum slope block displacement for the watertable at 6m (for later comparison).



11) Next raise the water level at the right-hand side of the model to 8 m above the slope toe; the water level on the left-hand side is maintained at the level of the slope toe.

bound pp=8e4 pygrad -1e4 range 21,23 -6 8

This may, for example, simulate a heavy rainfall event. Solving for the new fluid pressures:

cyc 500
save slfl3.sav

Plotting the groundwater flow vectors and the slope displacement vectors, check whether the flow affects the stability conditions.

plot bl vf yel	← flow vectors
plot bl pp red dis yel	← pore pressures and block displacements (at the same time)
plot bl sh yel	← this command plots the joint shear displacement magnitudes

Make note of (<u>record</u>) the maximum slope block displacement for the watertable at 8m. Ploting the joint shear it can be concluded that some shear slip now occurs but block failure still does not occur.

12) Again, raise the water level at the right-hand side of the model to 10 m above the slope toe.

bound pp=10e4 pygrad -1e4 range 21,23 -6 11

Solving now for the new fluid pressure condition.

cyc 16000
save slfl4.sav

Plotting the results, check to see whether the new groundwater conditions affects the stability conditions.

plot bl vf yel	← this command plots the flow vectors
plot bl dis yel	← this command plots the block displacements
plot bl pp yel	← this command plots the joint pore pressures
plot bl sh yel	← this command plots the flow vectors

Note the maximum slope block displacement for the watertable at 10m.

From this we see that by raising the water table level the slope becomes unstable. Significant block movements within the slope occurs when the fluid pressure in the joints increases (and the effective normal stress in the joints decreases) such that the limiting shear strength of the joints at the slope face is exceded. Thus, with the water level at 10 m, the slope fails, as indicated by the displaced rock wedge.